

## MK1 "Calcul formel" Maple

# TP7 : Développements limités

### But du TP7

Aujourd'hui nous allons utiliser Maple pour calculer des développements limités de fonctions et utiliser ces développements pour déterminer des asymptotes. Le terme "développement limité" est une spécialité française. Les anglo-saxons (en particulier Maple) préfèrent parler de développement de Taylor ("Taylor series expansion") ou de développement en série ("series expansion").

**Et surtout, n'oubliez pas de vous (et de me) poser des questions !**

## 1. Les développements de Taylor

Les développements de Taylor sont des cas particuliers de développements limités.

Rappel mathématique : Soit  $f$  une fonction sur un intervalle  $I$  de  $\mathbf{R}$ . Soit  $a \in I$ . Rappel la condition sur  $f$  pour qu'elle admette un développement de Taylor à l'ordre  $n$  et donner ce développement (formule de Taylor-Young).

Maple calcule les développements de Taylor avec la commande `taylor` (qui est un cas particulier de la commande `series`).

- > **restart**;
- > **?taylor**
- > **taylor(exp(-x), x, 4)**;  
$$1 - x + \frac{1}{2} x^2 - \frac{1}{6} x^3 + O(x^4) \quad (1)$$

La commande ci-dessus calcule le développement de Taylor de  $e^{(-x)}$  en  $x = 0$ , avec un "reste" en  $O(x^4)$ . Attention, il s'agit d'un  $O$ , et non pas d'un  $o$ .

Rappel mathématique :  $o$  et  $O$

$O(x^n)$  est une notation pour une fonction de  $x$  (appelons-la  $g$ ) telle que  $\frac{g(x)}{x^n}$  soit bornée au voisinage de  $0$ .

$o(x^n)$  est une notation pour une fonction de  $x$  (appelons-la  $h$ ) telle que  $\frac{h(x)}{x^n}$  tende

vers  $0$  quand  $x$  tend vers  $0$ .

Un  $O(x^n)$  est donc un  $o(x^{n-1})$ .

Dans l'exemple ci-dessus, on a donc obtenu un développement à l'ordre 3 au sens de votre cours de mathématiques. Plus généralement, si vous voulez obtenir un

développement à l'ordre  $k$  (au sens de votre cours de maths), il faut en général demander à Maple `taylor(f(x), x, k+1)`.

Cependant, attention ! La commande `taylor(f(x), x, k)` ne rend pas forcément une réponse en  $O(x^k)$ . En fait, Maple fera ses calculs intermédiaires à l'ordre Maple  $k$  mais rien ne garantit que le résultat final sera en  $O(x^k)$ . Regardez l'exemple :

> **taylor(tan(x), x, 6)**;  
$$x + \frac{1}{3} x^3 + \frac{2}{15} x^5 + O(x^6) \quad (2)$$

> **taylor(tan(x)/x, x, 6)**;  
$$1 + \frac{1}{3} x^2 + \frac{2}{15} x^4 + O(x^5) \quad (3)$$

On peut demander des développements ailleurs qu'en  $0$  :


> **taylor(exp(-x), x=1, 6)**;  
$$e^{(-1)} - e^{(-1)}(x-1) + \frac{1}{2} e^{(-1)}(x-1)^2 - \frac{1}{6} e^{(-1)}(x-1)^3 + \frac{1}{24} e^{(-1)}(x-1)^4 - \frac{1}{120} e^{(-1)}(x-1)^5 + O((x-1)^6) \quad (4)$$

Si on souhaite réutiliser la "partie principale" (= partie polynomiale) fournie par le développement, on procède avec la commande `convert(---, polynomial)` :

> **dev:=taylor(exp(-x), x, 4)**;  
$$dev := 1 - x + \frac{1}{2} x^2 - \frac{1}{6} x^3 + O(x^4) \quad (5)$$
  
> **P:=convert(dev, polynomial)**;  
$$P := 1 - x + \frac{1}{2} x^2 - \frac{1}{6} x^3 \quad (6)$$

On peut maintenant tracer  $e^{(-x)}$  et le polynôme  $P$  qui l'approche, au voisinage de  $0$  :

> **plot([exp(-x), P], x=-5..5, y=-5..5, color=[red, blue])**;



## 2. Les développements limités

Pour calculer des développements plus généraux que les développements de Taylor (développements généralisés, asymptotiques), on dispose de la commande *series*. Sa syntaxe est identique à celle de *taylor*.

Tout développement limité n'est pas forcément un développement de Taylor. Prenons la fonction définie par  $f(x) = x^4 \cos\left(\frac{1}{x^6}\right)$  pour  $x \neq 0$  et  $f(0) = 0$ . Elle a un

développement limité en 0 à l'ordre 3 qui est :  $f(x) = o(x^3)$  (pourquoi ?). Pourtant,  $f$  n'est pas dérivable deux fois en 0 (donc n'a pas de développement de Taylor), comme nous allons le voir.

> **f:=x->x^4\*cos(1/x^6);**  
**f(0):=0;**

$$f := x \rightarrow x^4 \cos\left(\frac{1}{x^6}\right)$$

$$f(0) := 0$$

(7)

Sa dérivée  $g$ , pour  $x \neq 0$ , est :

> **g:=D(f): g(x);**

$$4x^3 \cos\left(\frac{1}{x^6}\right) + \frac{6 \sin\left(\frac{1}{x^6}\right)}{x^3}$$

(8)

Calculons  $g(0)$ .

> **limit((f(x)-f(0))/(x-0), x=0);**

(9)

Donc  $g(0)=0$ . La fonction  $g$  n'est pas continue en 0 car :

> **limit(g(x), x=0);**

*undefined*

(10)

Donc  $g$  n'est pas dérivable en 0. Donc  $f$  n'est pas dérivable deux fois en 0.

Essayons d'obtenir le développement limité de  $f$  par Maple (on lui demande à l'ordre 4 pour obtenir un ordre "mathématique" 3).

> **taylor(f(x), x, 4);**

*Error, (in series/trig) unable to compute series*

> **series(f(x), x, 4);**

*Error, (in series/trig) unable to compute series*

Maple ne peut pas calculer ce développement. Rassurez-vous, en pratique, les fonctions utilisées sont très souvent suffisamment dérivables et vous pourrez utiliser Maple.

## Développement limité généralisé

Dans un développement limité, la partie principale (avant le 0 ou le O) est polynomiale. On peut définir des développements plus généraux. Dans un développement limité généralisé, on autorise des puissances négatives de  $x$  dans la partie principale. Un développement limité généralisé (en  $x=0$ ) est donc de la forme :

$$f(x) = \frac{1}{x^p} (a_0 + a_1 x + \dots + a_n x^n + o(x^n)) \text{ on et } p \text{ sont des entiers positifs.}$$

On peut calculer de tels développements dans Maple à l'aide de *series* (parfois *taylor* fonctionne également). Par exemple, pour obtenir le développement généralisé de

$\frac{1}{\tan(x)}$  à l'ordre (mathématique) 3 en 0 :

> **?series**

> **series(1/tan(x), x=0, 4);**

$$x^{(-1)} - \frac{1}{3} x + O(x^2) \quad (11)$$

## 3. Application : détermination d'asymptotes

Soit  $f$  une fonction et  $C$  sa courbe représentative dans le plan. On suppose que  $f$  est définie au voisinage de  $+\infty$  ou  $-\infty$ . Les développements limités généralisés permettent parfois de déterminer des asymptotes à l'infini de  $C$ .

### Asymptote horizontale ou oblique

Expliquez le résultat suivant : si on a le développement limité généralisé en  $+\infty$  ou  $-\infty$   $f(x) = ax + b + \frac{c}{x^p} + o\left(\frac{1}{x^p}\right)$  avec  $c \neq 0$  et  $p$  entier naturel non nul, alors la droite

d'équation  $y = ax + b$  est asymptote à  $C$  en  $+\infty$  ou  $-\infty$ .

De plus, le signe de  $\frac{c}{x^p}$  permet de déterminer la position de  $C$  par rapport à la droite.

Le cas particulier  $a = 0$  correspond à une asymptote horizontale.  
Voici un exemple :

> **f:=x->x\*arctan(x/(x-1));**

$$f := x \rightarrow x \arctan\left(\frac{x}{x-1}\right) \quad (12)$$

> **taylor(f(x), x=infinity, 3);**  
> **taylor(f(x), x=-infinity, 3);**

$$\frac{1}{4} \pi x + \frac{1}{2} + \frac{1}{4} \frac{1}{x} + O\left(\frac{1}{x^2}\right)$$

$$\frac{1}{4} \pi x + \frac{1}{2} + \frac{1}{4} \frac{1}{x} + O\left(\frac{1}{x^2}\right) \quad (13)$$


Donc la droite :

> **d:=Pi\*x/4+1/2;**

$$d := \frac{1}{4} \pi x + \frac{1}{2} \quad (14)$$

est asymptote à  $C$  en  $+\infty$  et en  $-\infty$ . De plus,  $C$  est au-dessus de la droite au voisinage de  $+\infty$  et en-dessous de la droite au voisinage de  $-\infty$ . On le voit sur un graphique :

> **plot([f(x), d], x=-5..5);**



### Courbe asymptote

Si on a le développement limité (généralisé ou asymptotique) en  $+\infty$  ou  $-\infty$

$f(x) = g(x) + \frac{c}{x^p} + o\left(\frac{1}{x^p}\right)$  avec  $c \neq 0$  et  $p$  entier naturel non nul, alors la courbe représentative  $C'$  de  $g$  est asymptote à  $C$  en  $+\infty$  ou  $-\infty$ . La position de  $C$  par rapport à  $C'$  est déterminée par le signe de  $\frac{c}{x^p}$ .

Si  $g$  est un polynôme de degré 1, on a une asymptote oblique (droite) ; si  $g$  est un polynôme de degré 2, on a une asymptote parabolique.

> **f:=x->x^3\*sin(1/x);**


$$f := x \rightarrow x^3 \sin\left(\frac{1}{x}\right) \quad (15)$$

> **series(f(x), x=infinity);**

$$x^2 - \frac{1}{6} + \frac{1}{120} \frac{1}{x^2} + O\left(\frac{1}{x^3}\right) \quad (16)$$

Donc la parabole d'équation  $y = x^2 - \frac{1}{6}$  est asymptote à la courbe  $C$  de  $f$  en  $+\infty$ .

> **plot([f(x), x^2-1/6], x=0..1, y=-1..1);**



Autre exemple (développement asymptotique) :

> **g:=x->ln(x^2+1)-1/x;**

$$g := x \rightarrow \ln(x^2 + 1) - \frac{1}{x} \quad (17)$$

> **series(g(x), x=infinity);**

$$2 \ln(x) - \frac{1}{x} + \frac{1}{x^2} - \frac{1}{2} \frac{1}{x^4} + O\left(\frac{1}{x^6}\right) \quad (18)$$

La courbe d'équation  $y = 2 \ln(x)$  est asymptote à  $C$  en  $+\infty$  et  $C$  est située sous son

asymptote au voisinage de  $+\infty$ .

```
> plot(Lg(x), 2*ln(x), x=0..15, y=-5..10, color=[red, blue]);
```

